



dataTaker DT500 Range Code Examples

Table of Contents

Introduction	2
Turning a digital output on and off at a set time	3
Gated input	3
DT5xx Gated input	3
Adaptive alarms	4
Sub second trigger	5
Series 3 loggers only	5
Logging machine downtime and reasons	5

Introduction

Example Programs are working examples produced with dataTaker 500 series loggers that can be downloaded and utilized as a starting point for your own projects or applications.

Hardware: DT50, DT51, DT500, DT600, DT505, DT605, DT515, and DT615 dataTaker data loggers and the CEM channel expansion module.

For more information visit [thermofisher.com/datataker](https://www.thermofisher.com/datataker)

Turning a digital output on and off at a set time

This program turns Digital Output 1 ON at 9:30AM and OFF at 4:30PM.

Note: Digital 1 is initially set to OFF in the 'immediate' schedule.

```
BEGIN
1DS0=0
RA10S HA
  1TK("Temp")
RZ10S HZ
  ALARM1(3ST("Hours")>9)AND
  ALARM2(2ST("Minutes")>30)"[1DS0=1]"
  ALARM3(3ST("Hours")>16)AND
  ALARM4(2ST("Minutes")>30)"[1DS0=0]"
END
```

Gated input

DT5xx Gated input

This is an example of the 1024 Hz gated input on a Series 3 DT5xx logger

- When 1HSC is pulled low counting starts and stops when open
- The counts are reported when 1DS goes high and 1HSC is reset
- The time is then calculated

Wire 1HSC and 1DS together. Then touch to ground.

Note: Event must be shorter than 2 second duration because the counter will wrap at 65535.

```
BEGIN
Y1=0,1"Sec" 'Dummy span for time units
P27=1 'Set up 32768 KHz input
RA1+E 'Report at end of event
  1HSC(R,=1CV) 'Report counts
  1CV(Y1)=1CV/32767 'Calculate Time in seconds
END
```

Adaptive alarms

A program example for automatically obtaining alarm setpoints for Alarms. Note comments in program.

```
'Reflow oven monitoring program.
'Portable loggers used on production ovens with varying process conditions.
'Logger must wait 15 minutes for temperatures to stabilise.
'Appropriate alarm setpoints are then calculated from last average and alarms enabled.
'Reset logger
RESET
'dataTaker schedule program
BEGIN
'Log statistical information every 15 minutes, do not return data
'Schedule runs only while digital input 3 is HIGH or OPEN from oven relay.
  RA15M:3W HA
    1TK("Section1",AV,NR)("Section1",MX,NR)("Section1",MN,NR)
    2TK("Section2",AV,NR)("Section2",MX,NR)("Section2",MN,NR)
    3TK("Section3",AV,NR)("Section3",MX,NR)("Section3",MN,NR)
    4TK("Section4",AV,NR)("Section4",MX,NR)("Section4",MN,NR)
    5TK("Section5",AV,NR)("Section5",MX,NR)("Section5",MN,NR)
'1 minute Counting schedule and loading of average temperature into CV
  RD1M:3W HD
    1CV(W)=1CV+1
    1TK("Section1",AV,=11CV,W)
    2TK("Section2",AV,=21CV,W)
    3TK("Section3",AV,=31CV,W)
    4TK("Section4",AV,=41CV,W)
    5TK("Section5",AV,=51CV,W)
'set statistical schedule rate
  RS30S
'set alarm schedule rate
  RZ5S HZ
'Reset alarm inhibit counter when digital input 3 is low
  IFR1(3DS<1)"[1CV(W)=0]"
'Enable thermal alarms after 15 minutes
  IFR2(1CV<15)"[HZ11 HZ12 HZ13 HZ14 HZ15]"
  IFR3(1CV>15)"[GZ11 GZ12 GZ13 GZ14 GZ15]"
'Calculate alarm setpoints from average temperature and load into channel variables
  IF5(1CV>15)"[12CV(W)=11CV-15 13CV(W)=11CV+15 22CV(W)=21CV-15 23CV(W)=21CV+15
    32CV(W)=31CV-15 33CV(W)=31CV+15 42CV(W)=41CV-15 43CV(W)=41CV+15 52CV(W)=51CV-15
    53CV(W)=51CV+15]"
'Alarm definitions
  ALARM11(1TK<>12CV,13CV)"Temperature Section 1 (?)^M^J"
  ALARM12(2TK<>22CV,23CV)"Temperature Section 2 (?)^M^J"
  ALARM13(3TK<>32CV,33CV)"Temperature Section 3 (?)^M^J"
  ALARM14(4TK<>42CV,43CV)"Temperature Section 4 (?)^M^J"
  ALARM15(5TK<>52CV,53CV)"Temperature Section 5 (?)^M^J"
END
'Enable datalogging
LOGON
'Start schedules
G
'Halt specific alarms on first loading of program to prevent confusion.
HZ11 HZ12 HZ13 HZ14 HZ15
```

Sub second trigger

Series 3 loggers only

A program example for triggering a DT5xx/6xx data logger faster than 1Hz. Requires 1 minor wiring link between Digital Input channel 1 and Digital Counter channel 1 Output terminals.

Refer to DT500 Series 3 User's Manual for detailed explanation of Parameter 27 and HSC Output modes.

This example provides a 0.5 sec nominal schedule trigger. Adjust divider setting and/or schedule trigger edge to obtain desired result. In addition, overall logger speed can be increased by adjusting P11 (mains freq) and P10 (ADC settling time) depending on individual requirements.

Note: The data obtained by this method cannot be viewed within DeLogger or DeView due to the sub 1 second speed. It must be unloaded as text and viewed in Excel or a similar spreadsheet.

```
'Reset logger
RESET
'Set input mode to Counter channel 1HSC
BEGIN
'Set output mode and divider for Counter channel 1HSC
1HSCO(3)=1024
P27=3
'Set Schedule A to trigger on Digital Input channel 1
  RA1E HA
    1V
    2V
    3V
    4V
END
'Enable data logging
LOGON

'Start schedules
G
```

Logging machine downtime and reasons

Program to log the reason and duration of machine downtime. Whenever the machine is down, a contact opens, and the operator presses one of 5 buttons which indicates the reason(s) for the downtime. When the machine is down, the DataTaker also turns on a warning lamp. Downtime reason buttons are connected as single ended analog states. Each downtime, the number and duration of all downtimes is logged, and the number and duration of downtimes due to each reason is logged.

```
'Reset the Datataker
RESET
\w5

'Set the Datataker clock to host computer clock time if required
D=\d
T=\t
BEGIN
'Initialize the individual downtime accumulators for each reason
```

```

1..5CV(W)=0          \ ' Number of downtimes for each reason button
6..10CV(W)=0        \ ' Duration of downtimes for each reason button
31..35CV(W)=0       \ ' Reason button ON/OFF flags

'Initialize the total downtime accumulators for all reasons
21CV(W)=0           \ ' Total number of downtimes
22CV(W)=0           \ ' Total duration of downtimes

'Pause while the Datataker completes initialisations
\w1

'Start measuring the downtime when machine contact opens (machine is turned off)
RA1+E
T(=23CV,W)          \ ' Save the time when machine goes down
5DS0=1              \ ' Turn the operator warning lamp on

'Check the reason buttons continuously while the machine is off and
'set the appropriate reason button ON/OFF flags
RZ:1W
IF1(1+AS(V)>0)"[31CV(W)=1]"
IF2(1-AS(V)>0)"[32CV(W)=1]"
IF3(2+AS(V)>0)"[33CV(W)=1]"
IF4(2-AS(V)>0)"[34CV(W)=1]"
IF5(3+AS(V)>0)"[35CV(W)=1]"

'Calculate and update downtimes when machine contact closes (machine turned back on)
RB1-E
T(=24CV,W)          \ ' Save the time when machine goes up
5DS0=0              \ ' Turn the operator warning lamp off
21CV(W)=21CV+1     \ ' Increment total downtimes counter
25CV(W)=24CV-23CV  \ ' Calculate the duration of this downtime
22CV(W)=22CV+25CV  \ ' and add to the total downtime

'Increment the downtime reason counters
1CV(W)=1CV+(31CV>0)
2CV(W)=2CV+(32CV>0)
3CV(W)=3CV+(33CV>0)
4CV(W)=4CV+(34CV>0)
5CV(W)=5CV+(35CV>0)

'Increment the downtime reason durations
6CV(W)=6CV+(31CV>0)*25CV
7CV(W)=7CV+(32CV>0)*25CV
8CV(W)=8CV+(33CV>0)*25CV
9CV(W)=9CV+(34CV>0)*25CV
10CV(W)=10CV+(35CV>0)*25CV

'Clear reason button ON/OFF flags
31..35CV(W)=0

'Log the data every hour
RC1H
1..5CV("Number",FF0)
6..10CV("Duration",FF2)
21CV("Total Number",FF0)
22CV("Total Duration",FF2)

END
LOGON
G

```

In Australia:

For customer service, call 1300-735-292

To email an order, ordersau@thermofisher.com

In New Zealand:

For customer service, call 0800-933-966

To email an order, ordersnz@thermofisher.com

Find out more at thermofisher.com/datataker

ThermoFisher
S C I E N T I F I C